NORDIC VOLCANOLOGICAL INSTITUTE 8101
UNIVERSITY OF ICELAND

Users guide to computer programs in BASIC for plotting geochemical data with a line printer

by

P. Thy & S.H. Sindrason

Nordic Volcanological Institute
University of Iceland
101 Reykjavik, Iceland

January 1981

NORDIC VOLCANOLOGICAL INSTITUTE 8101
UNIVERSITY OF ICELAND

Users guide to computer programs in BASIC for plotting geochemical data with a line printer

by

P. Thy & S.H. Sindrason

Nordic Volcanological Institute
University of Iceland
101 Reykjavik, Iceland

January 1981

1. Introduction

With the modern development in geochemical analytical techniques handling of large amounts of geochemical data has become an increasing necessity in geochemical and petrological research. Hand-plotting in orthogonal and trigonal diagrams is a troublesome and time consuming work. For these reasons low accuracy rutine plotting programs using a line printer has been developed to facilitate fast and realiable plotting of a large group or groups of data (e.g. Lumsden 1973, McHorne 1977). The present report describes plotting programs and a data handling system available at the Nordic Volcanological Institute. The programs are written in BASIC (CROMEMCO 1978) and implemented at a CROMEMCO microcomputer equipped with floppy disk storage facilities and a spinwriter printer.

Three main programs are available. A program for plotting in a orthogonal system (ORTPLOT), a program for trigonal plotting (TRIPLOT), and a program for printing histograms (HISTOGR). The trigonal plot program is a slightly modificated version of a program originally written by P. Rickwood at this institute. In addition a program for creation of input data files to the plotting programs (PLOTGEN) has been written. The main objective has been to allow the user not familiar with programming to handle and obtain useful results with the programs.

2. Data base

Creation of data bases is done with the program RECGEN (listed in appendix 1) where the geochemical data are read from the last lines in the program as data statements into a new created data record. In the present programs up to 14 variables are allowed. An example is given in Table 1. The first variable is a string (e.g. identification of the analyses in brackets ("BE 1.11")), the rest are numeric variables (up to 8 characters). The total number of analysis is only limited by the main memory of the computer.

The last lines in the new data base contain the string "END". The first line contains the total number of lines in the record, followed by a string variable read from the terminal (up to 40 characters), containing an explanatory text to the data base. A record is labelled 0,1,2,3,...,N. No attempt has so far been directed towards using segmented data records.

A data base, once created and stored, is convenient for various calculations (e.g. norms, ratios, etc.), tabulations of analysis, as well as data input for creation of input files to all the plotting programs. An example of a program simply reading the full content of a data base is given in Table 1. The user may be able to write her/his own programs for various forms of calculations and tabula-

Table 1

Listing of program for reading the content of a data base (TEST) and output from a RUN.

```
>>LOAD"RECREAD"

>>LIST

  100  FOR I=0 TO 70
  110  @"#"; : NEXT I   @
  120 @"PROGRAM TO READ CONTENT OF GEOCHEMICAL DATA RECORD."
  130 @"PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE."
  135 @"17.07.80  SHS & PT."
  140 @ : @ : @
  150 INPUT"GIVE NAME OF RECORD FILE TO READ?  ",S$
  160 INPUT"GIVE NUMBERS OF ELEMENTS IN ANALYSIS?  ",N
  170 @ : @ : @
  180 OPEN\1,120\S$
  190 DIM O$(49),N$(7),A(20)
  200 R=0
  210 GET\1,R\K,O$
  220 @K,O$
  230 R=1
  240 GET\1,R\N$(-1)
  250 @N$;"  ";
  260 IF N$="END"THEN GOTO 330
  270  FOR I=0 TO N-1
  280  GET\1,R,9+I*8\A(I)
  290  @USING"###.##",A(I);
  300  NEXT I
  310 R=R+1 : @
  320 GOTO 240
  330 @ : @ : @    FOR I=0 TO 70
  340  @"#"; : NEXT I : @
  350 CLOSE\1\
  360 STOP
```

```
>>RUN
################################################################################
PROGRAM TO READ CONTENT OF GEOCHEMICAL DATA RECORD.
PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE.
17.07.80  SHS & PT.


GIVE NAME OF RECORD FILE TO READ?  TEST
GIVE NUMBERS OF ELEMENTS IN ANALYSIS?  12


40                    TESTRUN
KE 1     47.92  2.35 14.62 10.29  0.17  6.29 12.37  2.33  0.91  0.39 81.30 75.80
KE 2     48.27  2.21 14.37  9.41  0.18  6.64 12.98  2.22  0.84  0.32 84.70 76.10
KE 3A    47.97  1.77 14.63 10.21  0.13  6.72 13.25  2.11  0.76  0.29 84.20 76.10
KE 3B    47.73  1.97 14.86  9.85  0.16  6.88 13.00  2.17  0.73  0.29 83.00 74.90
KE 4A    48.20  1.90 15.33  9.67  0.17  7.08 12.98  2.72  0.76  0.32 84.70 75.80
KE 4B    47.89  1.86 15.19  9.55  0.18  7.07 13.26  2.69  0.73  0.28 84.00 73.30
GR 1     48.02  1.67 15.61 10.35  0.18  6.39 11.06  2.90  0.95  0.29 83.10 70.70
GR 2     48.76  1.64 15.39 10.38  0.19  6.34 11.02  2.96  0.89  0.28 83.40 72.60
HI 1     48.78  1.97 14.65 11.37  0.20  5.92 10.80  2.57  0.99  0.32 80.20 74.20
HI 2F    47.74  2.24 14.39 12.55  0.22  6.12 10.75  2.70  0.99  0.31 79.90 73.60
HI 2C    48.96  1.91 14.98 11.76  0.21  6.22 10.82  2.52  0.96  0.32 79.80 72.70
BE 1     49.39  2.41 14.11 10.08  0.21  5.94 11.71  2.52  1.23  0.51 82.60 72.60
BE 2     50.05  2.40 13.75  9.60  0.17  5.80 11.51  2.52  1.20  0.47 82.80 73.60
BE 3     49.81  2.86 13.81 10.77  0.16  5.27 11.16  2.64  1.51  0.58 81.00 72.40
EN 3     46.87  2.70 14.17 10.65  0.20  6.71 12.83  3.01  0.91  0.52 85.70 74.20
EN 5     47.33  2.17 14.73 10.09  0.16  6.97 13.00  2.88  0.88  0.47 85.80 75.70
ST 1     46.51  2.40 15.11  9.53  0.17  7.26 13.32  3.04  0.09  0.41 86.10 74.50
ST 2     46.07  2.44 15.38 10.07  0.17  6.84 12.53  3.00  0.95  0.43 86.60 75.60
VE 20    46.20  3.08 15.11 13.07  0.22  5.57 10.35  3.75  0.74  0.36 76.50 66.60
VE 46    47.35  2.80 15.92 12.22  0.19  5.88 10.28  3.60  0.75  0.38 79.70 69.40
VE 49    46.94  3.07 14.89 13.57  0.24  5.14 10.38  3.95  0.83  0.42 77.70 70.80
VE 76    47.13  2.67 15.84 12.82  0.19  5.54 10.71  3.87  0.87  0.33 78.00 68.70
VE112    52.13  2.63 14.30 12.48  0.33  2.67  6.91  5.22  1.94  1.10 59.60 54.80
VE156    47.99  3.80 13.65 14.51  0.26  3.61  8.66  4.78  1.39  0.76 68.10 61.10
SU 19    47.21  2.51 15.58 11.57  0.23  5.84 10.58  4.04  0.67  0.36 80.30 75.00
SY 1     47.72  2.20 15.10 10.86  0.21  6.62 12.52  3.39  0.38  0.21 83.20 75.80
LA 1     47.05  3.96 12.78 14.45  0.24  5.28 10.56  4.41  0.61  0.60  0.00 62.60
LA 3     46.66  2.77 14.08 11.90  0.20  6.71 12.77  3.52  0.31  0.29 81.00 75.60
RE 1     46.35  2.68 14.25 11.78  0.19  7.18 12.63  3.37  0.33  0.31 80.40 74.70
RE 2     46.73  3.17 14.17 12.47  0.21  6.44 13.19  2.60  0.39  0.34 80.10 71.90
RE 4     46.27  2.77 14.21 12.01  0.20  7.00 13.38  2.51  0.33  0.30  0.00  0.00
RE 4     49.34  3.24 14.28 12.88  0.25  4.56  9.53  3.33  0.87  0.63  0.00  0.00
HA 1     46.59  3.91 12.86 14.27  0.24  5.10 10.69  2.88  0.61  0.48  0.00 65.30
HA 2     46.55  4.04 12.52 14.44  0.23  5.25 10.85  2.78  0.61  0.47  0.00 65.80
HA 3     46.75  3.92 12.91 14.36  0.25  5.13 10.83  3.22  0.61  0.57  0.00 56.50
HA 4A    46.87  4.31 12.99 14.34  0.24  4.60  9.63  3.15  0.79  0.92  0.00 58.90
HA 4B    47.53  4.21 12.71 14.29  0.25  4.62  9.58  3.48  0.77  0.76  0.00 57.10
HA 5     46.43  3.89 12.83 14.40  0.22  5.38 10.80  2.76  0.59  0.43 73.40 63.80
HA 6     46.92  4.19 13.52 14.21  0.24  4.72  9.70  3.15  0.73  0.62  0.00 59.10
END

################################################################################
***360 STOP***
```

tions using the examples given in Table 1. Consulting the few relevant pages in the BASIC manual on the OPEN, GET, PUT and CLOSE commands is, however, also recommended.

Note that all RUNs of programs described in this report must be terminated by a STOP command. If not make sure to write CLOSE\1\ on the terminal, otherwise your files can be seriously damaged.

3. Creation of input files for the plotting programs

The program PLOTGEN creates input files for the three plotting programs (Appendix 2). Data input for PLOTGEN is taken from the data base. The program is written for easy interactive communication with the user and is hence self-explanatory. The program asks for elements to store on the new file, two elements to ORTPLOT and one element to HISTOGR. Input to TRIPLOT is all elements in the original data base as this program contains options for deleting elements from the plotting (see description to TRIPLOT). The elements are identified by their numbers in the original record (1,2,3,...,N). The program contains options for calculations of mineral formula normalized to a chosen number of oxygen atoms. In this calculation the elements are identified by their atomic numbers (Fe^{3+} by zero). Only the oxides Na_2O (11), MgO (12), Al_2O_3 (13), SiO_2 (14), P_2O_5 (15), K_2O (19), CaO (20), TiO_2 (22), Cr_2O_3 (24), MnO (25), FeO (26) and Fe_2O_3 (0) are used in the present version of the program, but other elements can easily be enclosed by the user (from lines 940 and 1100 in the program as listed in Appendix 2). The user is advised to consider problems with missing values before a data base is created.

4. Description to TRIPLOT

The TRIPLOT program was original written by P. Rickwood. The present version contains only minor modifications (Appendix 3). The program plots in a fixed format triangular diagram (Fig. 1). The elements must be identified in



Fig. 1. Triangular plot using the program TRIPLOT. The figure shows the Mg:Fe:Na+K relations in basaltic glasses taken from the data base given in Table 1. By inserting the line 555 A(7)=A(7)+A(8) in PLOTGEN, the sum of Na_2O and K_2O is allocated to Na_2O's position in the input data for TRIPLOT. Figure reduced to 2/3.

the order entered from the data file, conveniently the atomic numbers (or other integer variables). The program contains options for deleting elements from the plots. Note that a 9 element data base will result in 86 different plots, if the numbers of elements are not reduced. It is hence recommended that the user carefully considers which

elements to plot before running the program. If the user wants to plot more complex variables this can easily be done by inserting the relevant lines in the PLOTGEN program (between lines 550 and 560 as listed in Appendix 2) and adding these variables to the original elements in the analysis. Plotting symbols are: * one analyses, 2-9 number of analysis, and # above or equal to 10.

A version of the program reading input from the last lines as data statements is available.

5. Description to ORTPLOT

The program ORTPLOT (Appendix 4) plots in a free format orthogonal diagram (Fig. 2). The program contains an option for the width of the x-axis (including frame to the plot) in printing characters and only limited by the total width of the paper (between 20 and 130). Furthermore max and min values of both axes can be given, but the program may be instructed to use the max and min values of the data set (0,0 causes default determinations of max and min). The resolution for the y-axis must always be given (see Fig. 2 plotted with a resolution of 0.50 for the y-axis). It is for these reasons suggested that the user carefully



Fig. 2. Orthogonal plot using the program ORTPLOT. The figure shows a common alkali-silica diagram of the basaltic glasses given in Table 1. A full printout from this run is given in Table 2. Figure reduced to 2/3.

considers the scaling of the plot before running the program.

The program contains subroutines for calculation of least-squares linear regression (Davis 1973:195-7). Single data pairs may be deleted from the linear regression calculation (up to 20). As the program makes use of interactive communication with the user the procedure is straight forward, but notice that if the reduced data set is plotted the original data input is lost. In order to restore the original data set it is necessary to RUN the program again. It should also be noted that if a limited part of the total range of the original data input is plotted, serious problems may occur in the linear regression calculations. Plotting symbols as for TRIPLOT.

A simplified version is available, not containing the linear regression subroutines, and reading data input from data statements as the last lines in the program.

6. Description to HISTOGR

The program HISTOGR (Appendix 5) plots data in a histogram (Fig. 3). The program contains options for resolution, max and min (0,0 causes default determination of max and min). As the program is fairly simple to use no further comments are needed. A version reading data from data statements is available.



Fig. 3. Histogram plotted using the program HISTOGR. The figure shows the forsterite content of olivine coexisting with basaltic glass (Table 1). Notice that the data base contains missing values (zero). These can easily be deleted by plotting from e.g. Fo_{40} to Fo_{90}. Figure reduced to 2/3.

7. Terminal session

A terminal session from a run with ORTPLOT is shown in Table 2 in order to illustrate to the user how to run his/her own data. The data base contains 39 microprobe

Table 2

Terminal session for plotting Fig. 2. Notice that element no. 8 has the position 7 in BASIC record.

```
>>LOAD"TESTRUN"

>>RUN
##########################################################################
PROGRAM TO CREATE GEOCHEMICAL DATA RECORD.
PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE  17.07.80.  SHS & PT.


DATA READ FROM LAST LINES IN PROGRAM.


GIVE NAME OF DATA RECORD?  TEST
GIVE NUMBERS OF ELEMENTS IN ANALYSIS?  12
TEXT TO DATA RECORD (MAX 40 CHARACTERS)?  TESTRUN
PROGRAM TERMINATED AND DATA STORED WITH THE FILE NAME: TEST


##########################################################################
***410 STOP***

>>LOAD"PLOTGEN"

>>555 A(7)=A(7)+A(8)

>>RUN
##########################################################################
PROGRAM TO CREATE INPUT FILE TO PLOTTING PROGRAMS.
PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE  17.07.80.  SHS & PT.



GIVE NAME OF ORIGINAL DATA RECORD?  TEST
GIVE NAME OF NEW DATA RECORD?  DATA
GIVE NUMBER OF ELEMENTS IN ORIGINAL DATA FILE?  12
DO YOU WANT TO CREATE DATA INPUT FILE TO TRIPLOT?  (YES OR NO):  NO
DO YOU WANT TO CREATE DATA INPUT FILE TO HISTOGR?  (YES OR NO):  NO
THE PROGRAM WILL CREATE DATA INPUT TO ORTPLOT.
GIVE THE NUMBER ON THE ELEMENTS TO STORE ON NEW RECORD.
FIRST ELEMENT IN ARRAY NO. 1 ETC.
X-ELEMENT?  1
Y-ELEMENT?  8
DO YOU WANT CALCULATION OF MINERAL FORMULA?  (YES OR NO):  NO
PROGRAM TERMINATED AND NEW DATA FILE STORED WITH THE NAME: DATA


##########################################################################
***790 STOP***
>>LOAD"ORTPLOT"
```

```
>>RUN
##########################################################################

ORTHOGONAL PLOTTING PROGRAM.      VER. 2.


NORDIC VOLCANOLOGICAL INSTITUTE. 17.07.80. SHS & PT.


GIVE DATA FILE:  DATA
LINE WIOTH:   70
Y-AXIS:  ALKALIES
Y MAX, Y MIN, RESOLUTION (0,0,N CAUSES DEFAULT):  8.5,0,0.5
X-AXIS:  SIO2
X MAX, X MIN (0,0 CAUSES DEFAULT):  53.5,45
LINEAR REGRESSION? (YES OR NO):  NO



TESTRUN
TOTAL NUMBER OF ANALYSIS:                39
X-AXIS:            SIO2                  Y-AXIS:          ALKALIES
```

(Output shown in Fig. 2)

```
TOTAL NUMBERS IN PLOT:                   39


DO YOU WANT TO TERMINATE THE PROGRAM?  (YES OR NO):  YES
***2350 STOP***
```

analysis of basaltic glasses from Iceland together with co-existing composition of olivine (Fo) and plagioclase (An) (Table 1). Results from the other runs were already shown in Figs. 1 and 3.

8. Concluding remarks

The program package described in this report forms an easy and fast method for storage and plotting of a large number of geochemical data. The programs, based on interactive communication with the user, needs only very little programming background. Hardware requirements are a microcomputer, floppy disk storage facilities and a line printer or other printer types, which today are found in most geochemical and petrological departments.

Further needs for development of the program package may be pointed out, including subroutines for normative calculations, more sophisticated mineral calculations and segmentated data records. The latter will allow several data groups to be plotted on the same diagram with different symbols.

References

CROMEMCO 1978: 16K extended BASIC. Instruction Manual. CROMEMCO.

Davis, J.C. 1973: Statistics and Data Analysis in Geology. Wiley, New York.

Lumsden, D.N. 1973: TRI: A FORTRAN subroutine to plot points on a triangular diagram. Geol. Soc. Am. Bull. 84, 1765-1768.

McHone, J.G. 1977: Triplot: An APL program for plotting triangular diagrams. Computers & Geosciences 3, 633-635.

Appendix 1. Listing of RECGEN

```
>>LOAD"RECGEN"

>>LIST

  100  FOR I=0 TO 70
  110  @"#"; : NEXT I : @
  120 @"PROGRAM TO CREATE GEOCHEMICAL DATA RECORD."
  130 @"PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE  17.07.80.  SHS & PT.
  140 @ : @ : @
  150 @"DATA READ FROM LAST LINES IN PROGRAM."
  160 @ : @ : @
  170 INPUT"GIVE NAME OF DATA RECORD?  ",S$
  180 CREATE S$
  190 INPUT"GIVE NUMBERS OF ELEMENTS IN ANALYSIS?  ",N
  200 DIM G$(40),N$(7),A(20)
  210 INPUT"TEXT TO DATA RECORD (MAX 40 CHARACTERS)?  ",O$
  220 OPEN\1,120\S$
  230 N$="XXX"
  240 R=0
  250 PUT\1,R\K,O$
  260 IF N$="END"THEN GOTO 370
  270 R=1
  280 READ N$
  290 IF N$="END"THEN PUT\1,R\N$(-1)   GOTO 240
  300 PUT\1,R\N$(-1)
  310  FOR I=0 TO N-1
  320  READ A(I)
  330  PUT\1,R,9+I*8\A(I)
  340  NEXT I
  350 R=R+1 : K=R
  360 GOTO 280
  370 @"PROGRAM TERMINATED AND DATA STORED WITH THE FILE NAME: ";S$
  380 @ : @ : @ :  FOR I=0 TO 70
  390  @"#"; : NEXT I : @
  400 CLOSE\1\
  410 STOP
  420 DATA"BE 1.11",0.16,0.66,19.99,45.31,4.91,17.03,0.21,12.36,0.0
  430 DATA"BE 1.12",0.16,0.69,19.91,45.66,4.51,17.46,0.23,12.1,0.0
  440 DATA"BE 1.13",0.22,0.55,19.69,46.36,5.36,15.53,0.28,13.41,0.0
  450 DATA"BE 1.24",0.11,0.64,19.95,42.35,7.41,16.93,0.34,12.09,0.0
  460 DATA"END"
```

Appendix 2. Listing of PLOTGEN

```
>>LOAD"PLOTGEN"

>>LIST

  100  FOR I=0 TO 70
  110  @"#"; : NEXT I : @
  120 DIM P$(40),A(20),B(20)
  130 @"PROGRAM TO CREATE INPUT FILE TO PLOTTING PROGRAMS."
  140 @"PLOTTING SYSTEM VER. 2. NORDIC VOLCANOLOGICAL INSTITUTE  17.07.80.  SHS & PT."
  150 @ : @ : @
  160 W$="NO" : V$="NO" : Z$="NO"
  170 INPUT"GIVE NAME OF ORIGINAL DATA RECORD?  ",S$
  180 OPEN\1,120\S$
  190 GET\1,0\M,P$
  200 CLOSE\1\
  210 INPUT"GIVE NAME OF NEW DATA RECORD?  ",B$
  220 CREATE B$
  230 OPEN\1,120\B$
  240 PUT\1,0\M,P$
  250 CLOSE\1\
  260 INPUT"GIVE NUMBER OF ELEMENTS IN ORIGINAL DATA FILE?  ",N
  270 INPUT"DO YOU WANT TO CREATE DATA INPUT FILE TO TRIPLOT?  (YES OR NO):  ",W$
  280 IF W$="YES"THEN@"THE PROGRAM WILL CREATE DATA INPUT TO TRIPLOT." : GOTO 400
  290 INPUT"DO YOU WANT TO CREATE DATA INPUT FILE TO HISTOGR?  (YES OR NO):  ",V$
  300 IF V$="YES"THEN@"THE PROGRAM WILL CREATE DATA INPUT TO HISTOGR." : GOTO 370
  310 @"THE PROGRAM WILL CREATE DATA INPUT TO ORTPLOT."
  320 @"GIVE THE NUMBER ON THE ELEMENTS TO STORE ON NEW RECORD."
  330 @"FIRST ELEMENT IN ARRAY NO. 1 ETC."
  340 INPUT"X-ELEMENT?  ",A
  350 INPUT"Y-ELEMENT?  ",B
  360 IF V$="NO"OR W$="NO"THEN GOTO 400
  370 @"GIVE THE NUMBER ON THE ELEMENT TO STORE ON THE NEW RECORD."
  380 @"FIRST ELEMENT IN ARRAY NO. 1 ETC."
  390 INPUT"ELEMENT?  ",A
  400 R=1
  410 INPUT"DO YOU WANT CALCULATION OF MINERAL FORMULA?  (YES OR NO):  ",Z$
  420 IF Z$="NO"THEN GOTO 490
  430 @"IDENTIFY WT% OXIDES BY ATOMIC NUMBERS (FE3+=0) IN THE SAME ORDER"
  440 @"AS THE ORIGINAL DATA RECORD (ONLY 11,12,13,14,15,19,20,22,24,25,26)."
  450  FOR I=0 TO N-1
  460  @"ELEMENT NO. ";I+1; : INPUT B(I)
  470  NEXT I
  480 INPUT"NUMBER OF OXYGENS?  ",T1
  490 OPEN\1,120\S$
  500 IF R=M THEN GOTO 750
  510  FOR I=0 TO N-1
  520  GET\1,R,9+I*8\A(I)
  530  NEXT I
  540 CLOSE\1\
  550 IF Z$="YES"THEN GOSUB 800
  560 IF W$="YES"THEN GOTO 630
  570 IF V$="YES"THEN GOTO 700
  580 OPEN\1,120\B$
  590 PUT\1,R\A(A-1),A(B-1)
  600 R=R+1
  610 CLOSE\1\
  620 GOTO 490
  630 OPEN\1,120\B$
  640  FOR I=0 TO N-1
  650  PUT\1,R,9+I*8\A(I)
  660  NEXT I
  670 R=R+1
  680 CLOSE\1\
  690 GOTO 490
  700 OPEN\1,120\B$
  710 PUT\1,R\A(A-1)
  720 R=R+1
  730 CLOSE\1\
  740 GOTO 490
  750 @"PROGRAM TERMINATED AND NEW DATA FILE STORED WITH THE NAME: ";B$
  760 @ : @ : @    FOR I=0 TO 70
  770  @"#"; : NEXT I : @
  780 CLOSE\1\
```

```
 800 REM SUBR TO CALCULATE MINERAL FORMULA
 810 T=0
 820  FOR I=0 TO N-1
 830  IF B(I)=14 THEN A(I)=2*A(I)/60.0852 : GOTO 950
 840  IF B(I)=22 THEN A(I)=2*A(I)/79.8988 : GOTO 950
 850  IF B(I)=13 THEN A(I)=3*A(I)/101.9612 : GOTO 950
 860  IF B(I)=24 THEN A(I)=3*A(I)/151.9902 : GOTO 950
 870  IF B(I)=0 THEN A(I)=3*A(I)/159.6922 : GOTO 950
 880  IF B(I)=26 THEN A(I)=A(I)/71.8464   GOTO 950
 890  IF B(I)=25 THEN A(I)=A(I)/70.9374 : GOTO 950
 900  IF B(I)=12 THEN A(I)=A(I)/40.3114 : GOTO 950
 910  IF B(I)=20 THEN A(I)=A(I)/56.0796 : GOTO 950
 920  IF B(I)=11 THEN A(I)=A(I)/61.979 : GOTO 950
 930  IF B(I)=19 THEN A(I)=A(I)/94.2034 : GOTO 950
 940  IF B(I)=15 THEN A(I)=5*A(I)/141.9446 : GOTO 950
 950  T=T+A(I)
 960  NEXT I
 970 T=T1/T : T2=0
 980  FOR I=0 TO N-1
 990  IF B(I)=14 THEN A(I)=A(I)*T/2 : GOTO 1110
1000  IF B(I)=22 THEN A(I)=A(I)*T/2 : GOTO 1110
1010  IF B(I)=13 THEN A(I)=A(I)*T*2/3 : GOTO 1110
1020  IF B(I)=24 THEN A(I)=A(I)*T*2/3 : GOTO 1110
1030  IF B(I)=0 THEN A(I)=A(I)*T*2/3 : GOTO 1110
1040  IF B(I)=26 THEN A(I)=A(I)*T : GOTO 1110
1050  IF B(I)=25 THEN A(I)=A(I)*T : GOTO 1110
1060  IF B(I)=12 THEN A(I)=A(I)*T : GOTO 1110
1070  IF B(I)=20 THEN A(I)=A(I)*T : GOTO 1110
1080  IF B(I)=11 THEN A(I)=A(I)*T*2   GOTO 1110
1090  IF B(I)=19 THEN A(I)=A(I)*T*2   GOTO 1110
1100  IF B(I)=15 THEN A(I)=A(I)*T*2/5 : GOTO 1110
1110  T2=T2+A(I)
1120  NEXT I
1130 RETURN
```

Appendix 3. Listing of TRIPLOT

```
>>LOAD"TRIPLOT"

>>LIST

  100 SFMODE
  110  FOR I=0 TO 72
  120  @"#"; : NEXT I : @
  130 @"TRIANGULAR PLOTTING PROGRAM     VER. 2  -."
  140 @ : @"NORDIC VOLCANOLOGICAL INSTITUTE, REYKJAVIK."
  150 @"PROGRAM WRITTEN BY PETER RICKWOOD. PT-80/7/3."
  160 @ : @ : @
  170  FOR I=0 TO 72
  180  @"-"; : NEXT I : @
  190 @ : @"PATIENCE - CLEARING ARRAYS TAKES 20 SECS."
  200 @
  210 DIM P$(40)
  220 DIM Z3(12),F6(12)
  230 DIM P(30,60),E1(12),E2(12),E(92)
  240  FOR I=0 TO 12
  250  E1(I)=0 : E2(I)=0
  260  NEXT I
  270  FOR I=0 TO 92
  280  E(I)=0
  290  NEXT I
  300  FOR J=0 TO 30
  310   FOR I=0 TO 60
  320   P(J,I)=0
  330   NEXT I
  340  NEXT J
  350 INPUT"GIVE NAME OF DATA FILE?  ",S$
  360 OPEN\1,120\S$
  370 GET\1,0\N1,P$
  380 N1=N1-1
  390 N5=N1
  400 @ : INPUT"HOW MANY ELEMENTS PER ANALYSIS  ",N2
  410 N3=N2 : N4=0
  420 @ : @"SPECIFY"
  430 IF N3<=12 THEN@N3   N3=0 : GOTO 450
  440 @"   12   ";
  450 @"ATOMIC NUMBERS OF ELEMENTS IN ORDER ENTERED"
  460  FOR I=0 TO N2-1
  470  INPUT E2(I)
  480  NEXT I
  490 @
  500  FOR I=1 TO 12
  510  E(12*N4+I)=E2(I-1)
  520  NEXT I
  530 IF N3>12 THEN N3=N3-12
  540 N4=N4+1
  550 IF N3<>0 THEN GOTO 420
  570 REM TRIANGULAR PLOT
  580 Q=N2*(N2-1)*(N2-2)/6
  590 @ : @"THE NUMBER OF TRIANGULAR GRAPHS TO PLOT IS ",Q
  600 GOSUB 1710
  610 Q=N3*(N3-1)*(N3-2)/6
  620 @ : @"NOW THE NUMBER OF TRIANGULAR GRAPHS TO PLOT IS ",Q
  630 @ : @ : @ :  FOR I=0 TO 72
  640  @"~"; : NEXT I : @
  650 I6=0
  660 Q=0 : @ : @
  670  FOR M1=1 TO N2-2
  680  X=M1 : GOSUB 1980
  690  IF X1>0 THEN GOTO 1690
  700   FOR M2=M1+1 TO N2-1
  710   X=M2 : GOSUB 1980
  720   IF X1>0 THEN GOTO 1680
  730    FOR M3=M2+1 TO N2
  740    X=M3 : GOSUB 1980
  750    IF X1>0 THEN GOTO 1670
  760    Q=Q+1
  770     FOR I=1 TO N1
  780      FOR I1=1 TO N2
  790      GET\1.I.1+I1*8\09
```

```
 810      IF I1<>M1 THEN IF I1<>M2 THEN IF I1<>M3 THEN GOTO 860
 820      IF O9=999 THEN O9=0
 830      IF I1=M1 THEN A=O9   GOTO 860
 840      IF I1=M2 THEN B=O9 : GOTO 860
 850      IF I1=M3 THEN C=O9
 860      NEXT I1
 870     IF I>1 THEN GOTO 980
 880     IF A=0 THEN A1=0   GOTO 900
 890     A1=0
 900     IF B=0 THEN B1=0   GOTO 920
 910     B1=0
 920     IF C=0 THEN C1=0   GOTO 940
 930     C1=0
 940     M9=A1
 950     IF M9<B1 THEN M9=B1
 960     IF M9<C1 THEN M9=C1
 970     A1=M9-A1 : B1=M9-B1   C1=M9-C1
 980     A=A*10^A1 : B=B*10^B1 : C=C*10^C1
 990     T=A+B+C
1000     IF T=0 THEN GOTO 1070
1010     X=A/T*100
1020     Y=B/T*100
1030     Z=C/T*100
1040     Y=INT(Z*0.6+X*0.3+0.5)
1050     X=INT(30-X/100*30+0.5)
1060     GOSUB 2080
1070     NEXT I
1080     FOR J=0 TO 44
1090     @" ";
1100     NEXT J
1110    @E(M1); : @"/";   @E(M2);   @"/";   @E(M3);
1120    @"      GRAPH NO. ";Q : @
1130    @P$ : @
1140     FOR J=0 TO 34
1150     @" ";
1160     NEXT J
1170    @E(M1);
1180    IF A1<>0 THEN@"*E",A1;
1190    @
1200    J2=0
1210     FOR J=0 TO 30
1220     @"     ";
1230     Y1=30-J
1240     Y2=J+30
1250     IF J=30 THEN GOTO 1290
1260      FOR K=0 TO 29-J
1270      @" ";
1280      NEXT K
1290      FOR K=30-J TO 30+J
1300      IF P(J,K)<>0 THEN GOTO 1420
1310      IF J=0 THEN IF K=30 THEN IF P(0,30)=0 THEN@".";   GOTO 1450
1320      IF J<>30 THEN GOTO 1370
1330      IF K=60 THEN@"."; : GOTO 1450
1340      IF K=0 THEN GOTO 1360
1350      IF INT(K/6)=K/6 THEN@"\"; : GOTO 1440
1360      @"."; : GOTO 1440
1370      IF K=30-J THEN GOTO 1400
1380      IF K=30+J THEN GOTO 1400
1390      @" "; : GOTO 1440
1400      IF J2=3 THEN@"~";   GOTO 1430
1410      @"."; : GOTO 1430
1420      I9=P(J,K) : GOSUB 2220
1430      IF K=Y2 THEN GOTO 1450
1440      NEXT K
1450     @
1460     J2=J2+1 : IF J2=4 THEN J2=1
1470     NEXT J
1480    @"     ";
1490    @E(M2); : A1=62
1500    IF B1<>0 THEN@"*E";B1;
1510    A1=A1-4
1520     FOR L=0 TO A1
1530     @" ";
1540     NEXT L
1550    @E(M3);
1560    IF C1<>0 THEN@"*E";C1;
```

```
1580     FOR K=1 TO X
1590     @
1600     NEXT K
1610     FOR J=0 TO 30
1620      FOR K=30-J TO 30+J
1630      P(J,K)=0
1640      NEXT K
1650     NEXT J
1660    I6=0
1670    NEXT M3
1680   NEXT M2
1690  NEXT M1
1700 GOTO 2030
1710 REM SUBROUTINE TO REDUCE NO. OF ELEMENTS TO PLOT
1720  FOR I=0 TO 12
1730  E1(I)=0
1740  E2(I)=0
1750  NEXT I
1760 IF Q=20 THEN GOTO 660
1770 @"TRY AND REDUCE THIS NUMBER WHEN ANSWERING THE FOLLOWING QUESTIONS"
1780 @ : @"SPECIFY ATOMIC SYMBOLS OF UP TO 12 ELEMENTS THAT";
1790 @" YOU DO NOT WHISH TO PLOT"
1800  FOR I=0 TO N2-1
1810  INPUT E2(I)
1820  NEXT I
1830 @
1840  FOR I=0 TO 11
1850  IF E2(I)=0 THEN GOTO 1870
1860  NEXT I
1870 N3=N2-I
1880 IF N3>2 THEN GOTO 1950
1890 @"YOU HAVE DELETED TOO MANY ELEMENTS"
1900  FOR I=0 TO 11
1910  E1(I)=0
1920  E2(I)=0
1930  NEXT I
1940 GOTO 1780
1950 RETURN
1960 @ : @
1970 GOTO 560
1980 X1=0
1990  FOR I=0 TO 11
2000  IF E(X)=E2(I)THEN X1=X1+1
2010  NEXT I
2020 RETURN
2030 @"PROGRAM TERMINATED."
2040 CLOSE\1\
2050  FOR I=0 TO 70
2060  @"#"; : NEXT I : @
2070 STOP
2080 J=X
2090 K=Y
2100 P(J,K)=P(J,K)+1
2110 RETURN
2120 REM PLOTTING SUBROUTINE
2130  FOR J4=1 TO Y3
2140  @" ";
2150  NEXT J4
2160 Y1=Y1+Y3
2170 RETURN
2180 IF J2=3 THEN@"-";   GOTO 2210
2190 @".";
2200 Y1=Y1+1
2210 RETURN
2220 IF I9=1 THEN@"*";   GOTO 2320
2230 IF I9=2 THEN@"2";   GOTO 2320
2240 IF I9=3 THEN@"3";   GOTO 2320
2250 IF I9=4 THEN@"4";   GOTO 2320
2260 IF I9=5 THEN@"5";   GOTO 2320
2270 IF I9=6 THEN@"6";   GOTO 2320
2280 IF I9=7 THEN@"7";   GOTO 2320
2290 IF I9=8 THEN@"8";   GOTO 2320
2300 IF I9=9 THEN@"9";   GOTO 2320
2310 IF I9>=10 THEN@"#"; : GOTO 2320
2320 Y1=Y1+1
2330 RETURN
```

```
2340 IF Y1=P(J,0)THEN Y3=Y1 : Y1=0 : GOTO 2420
2350 IF Y1>30-J THEN GOTO 2400
2360 IF J=30 THEN IF Y1=0 THEN GOTO 2400
2370 Y3=Y1 : Y1=0
2380 GOSUB 2120
2390 GOSUB 2180
2400 Y3=P(J,K)-Y1
2410 IF J=30 THEN IF P(J,K)=-1 THEN Y3=0
2420 IF Y3=0 THEN GOTO 2480
2430 IF Y3=-1 THEN@" "; : Y3=0   Y1=Y1-1   GOTO 2480
2440 IF Y3=-2 THEN@" "; : Y3=0 : Y1=Y1-2   GOTO 2480
2450 IF J=30 THEN GOSUB 2510
2460 IF J=30 THEN GOTO 2480
2470 GOSUB 2120
2480 GOSUB 2220
2490 I9=1
2500 RETURN
2510  FOR J4=1 TO Y3
2520  IF Y1=0 THEN GOTO 2550
2530  IF Y1=60 THEN GOTO 2550
2540  IF INT(Y1/6)=Y1/6 THEN@"\";   GOTO 2560
2550  @".";
2560  Y1=Y1+1
2570  NEXT J4
2580 RETURN
```

Appendix 4. Listing of ORTPLOT

```
>>LOAO"ORTPLOT"

>>LIST

  100 REM ORTHOGONAL PLOTTING PROGRAM DATA READ FROM RECORD FILE
  110  FOR I=0 TO 70
  120  @"#"; : NEXT I : @
  130 @ : @ : @"ORTHOGONAL PLOTTING PROGRAM.      VER. 2."
  140 @ : @ : @
  150 @"NORDIC VOLCANOLOGICAL INSTITUTE. 17.07.80. SHS & PT."   @ : @ : @
  160 INPUT"GIVE DATA FILE:  ",S$
  170 SET 0,130
  180 DIM T(150),J(150)
  190 DIM G1(150),J1(150)
  200 DIM P$(40),K$(15),L$(15)
  220 OPEN\1,120\S$
  230 GET\1,0\N,P$
  240 N=N-2
  250  FOR I=0 TO N
  260  GET\1,I+1\J(I),T(I)
  270  NEXT I
  280 CLOSE\1\
  290 GOSUB 380
  300 @ : @"TOTAL NUMBERS IN PLOT:  ",X9   @
  310 IF U$="YES"THEN GOSUB 1520
  320  FOR I=0 TO 9
  330  @ : NEXT I
  340 IF Z$="YES"THEN GOTO 370
  350 INPUT"DO YOU WANT TO TERMINATE THE PROGRAM?  (YES OR NO):  ",Q$
  360 IF Q$="YES"THEN GOTO 2350
  370 Z$="XXX" : GOTO 290
  380 REM PLOT RUTINE T-ARRAY=Y AXIS, J-ARRAY=X AXIS
  390 INPUT"LINE WIDTH:   ",F
  400 X9=0
  410 IF F>130 OR F<20 THEN@"CHOSE LINE WIDTH BETWEEN 20 AND 130!" : GOTO 390
  420 IF INT(F/10)<>(F/10)THEN F=10*(INT(F/10)) : @"LINE WIDTH CHANGED TO:  ",F
  430 F=F-3
  440 INPUT"Y-AXIS:  ",K$
  450 INPUT"Y MAX, Y MIN, RESOLUTION (0,0,N CAUSES DEFAULT):  ",T2,T1,R
  460 IF T2=0 AND T1=0 THEN GOSUB 630
  470 X8=(T2-T1)/R
  480 IF INT(X8)<>X8 THEN T2=(INT(X8)*R)+R+T1 : @"Y MAX CHANGED TO:  ",T2
  490 INPUT"X-AXIS:  ",L$
  500 INPUT"X MAX, X MIN (0,0 CAUSES DEFAULT):  ",Y2,Y1
  510 IF Y2=0 AND Y1=0 THEN GOSUB 700
  520 IF Y2-Y1<=0 THEN GOTO 2350
  530 INPUT"LINEAR REGRESSION? (YES OR NO):  ",U$
  540  FOR I=0 TO 9
  550  @ : NEXT I
  560 @P$ : @
  570 IF Z$="YES"THEN@"DATA FILE NOW DELETED WITH ";W;" ANALYSIS TO ";N+1;" ." :
  580 @"TOTAL NUMBER OF ANALYSIS:   ",N+1 : @
  590 @"X-AXIS:  ",L$,"  Y-AXIS:  ",K$                                  @ : GOTO 500
  600 @ : @ : @
  610 GOSUB 770
  620 RETURN
  630 REM SUBR MAX, MIN Y-AXIS
  640 T2=T(0) : T1=T(0)
  650  FOR I=0 TO N
  660  T2=MAX(T2,T(I)) : T1=MIN(T1,T(I))
  670  NEXT I
  680 @"Y MAX=";T2,"Y MIN=";T1
  690 RETURN
  700 REM SUBR MAX, MIN X-AXIS
  710 Y2=J(0) : Y1=J(0)
  720  FOR I=0 TO N
  730  Y2=MAX(Y2,J(I)) : Y1=MIN(Y1,J(I))
  740  NEXT I
  750 @"X MAX=";Y2,"X MIN=";Y1
  760 RETURN
  770 REM SUBR PLOT Y
  780  FOR I=T1 TO T2 STEP R
  790   FOR L=0 TO 150
  800   J1(L)=0 : NEXT L
```

```
 810  L1=0
 820  GOSUB 1230
 830   FOR L=0 TO N
 840   IF T(L)<=(I+R/2)THEN IF T(L)>(I-R/2)THEN GOSUB 1330
 850   NEXT L
 860  GOSUB 900
 870  NEXT I
 880 GOSUB 1370
 890 RETURN
 900 REM SUBR PLOT X
 910 IF L1=0 THEN 1210
 920 L2=0
 930 J1(L2)=J(G1(L2))
 940 L2=L2+1 : IF L2<L1 THEN GOTO 930
 950  FOR K=0 TO F-7
 960  G1(K)=0
 970  NEXT K
 980 G2=0
 990 L2=0
1000 IF L1=1 THEN GOTO 1060
1010 REM INCR
1020 IF J1(L2+1)>=J1(L2)THEN 1040
1030 G2=J1(L2+1) : J1(L2+1)=J1(L2) : J1(L2)=G2   G2=99
1040 L2=L2+1 : IF L2<L1-1 THEN GOTO 1010
1050 IF G2=99 THEN 980
1060  FOR K=0 TO L1-1
1070  J1(K)=(J1(K)-Y1)*(F-7)/(Y2-Y1)
1080  NEXT K
1090  FOR K=0 TO F-7
1100   FOR L=0 TO L1-1
1110   IF J1(L)>=(K-0.5)THEN IF J1(L)<(K+0.5)THEN G1(K)=G1(K)+1
1120   NEXT L
1130  NEXT K
1140  FOR K=0 TO F-7
1150  IF G1(K)>=10 THEN@"#";
1160  IF G1(K)=0 THEN@" ";
1170  IF G1(K)>1 AND G1(K)<10 THEN@G1(K);
1180  IF G1(K)=1 THEN@"*":
1190  X9=X9+G1(K)
1200  NEXT K
1210 @TAB(F+3);"!"
1220 RETURN
1230 REM SUBR PRINT Y-VALUES
1240 IF I<>T1 THEN GOTO 1300
1250 @TAB(5);
1260 X8=((F-7)/2)-7 : X7=(Y2+Y1)/2
1270 @USING"-###.##",Y1;
1280 @SPC(X8); : @USING"-###.##",X7;   @SPC(X8);   @USING"-###.##",Y2
1290 GOSUB 1370
1300 @USING"-###.##",I;
1310 PRINT" !";
1320 RETURN
1330 REM SUBR L
1340 G1(L1)=L
1350 L1=L1+1
1360 RETURN
1370 REM SUBR FRAME
1380 B2=0
1390  FOR L3=0 TO 8
1400  @" "; : NEXT L3
1410 B1=(F-7)/11
1420 @"!";
1430 B2=B2+1
1440  FOR L3=0 TO B1-1
1450  @"-";
1460  B2=B2+1
1470  NEXT L3
1480 IF B2<F-7 THEN GOTO 1420
1490 IF B2=F-7 THEN@"!"
1500 IF B2>F-7 THEN GOTO 1510
1510 RETURN
1520 REM SUBR FOR LINEAR REGRESSION
1530 REM LEAST-SQUARES LINEAR REGRESSION CALCULATED FROM FORMULA
1540 REM GIVEN IN DAVIS: "STATISTICS AND DATA ANALYSIS IN GEOLOGY",
1550 REM WILEY 1973, FORMULA 5.5,5.9,5.10,5.14,5.16.
```

```
1560 X9=N+1
1570 DIM A(1,X9-1),B(X9-1)
1580  FOR L=0 TO X9-1
1590  A(0,L)=0 : A(1,L)=0
1600  B(L)=0
1610  NEXT L
1620 X=0
1630  FOR L=0 TO N
1640  IF T(L)>T2 OR T(L)<T1 THEN GOTO 1670
1650  IF J(L)>Y2 OR J(L)<Y1 THEN GOTO 1670
1660  A(0,X)=J(L) : A(1,X)=T(L)
1670  X=X+1
1680  NEXT L
1690 X1=0 : S3=0 : N1=0 : Z1=0   Z2=0   83=0   C1=0   84=0
1700 IF X9=0 THEN GOTO 1990
1710  FOR I=0 TO X9-1
1720  X1=X1+A(0,I)
1730  S3=S3+A(1,I)
1740  Z1=Z1+(A(0,I)*A(1,I))
1750  Z2=Z2+(A(0,I)*A(0,I))
1760  C1=C1+(A(1,I)*A(1,I))
1770  NEXT I
1780 Z3=Z1-((X1*S3)/X9)
1790 Z4=Z2-((X1*X1)/X9)
1800 IF Z4=0 THEN GOTO 1990
1810 S5=Z3/Z4
1820 S4=(S3/X9)-((S5*X1)/X9)
1830  FOR I=0 TO X9-1
1840  B(I)=S4+(S5*A(0,I))
1850  83=B3+B(I)
1860  B4=B4+(B(I)*B(I))
1870  NEXT I
1880 S1=C1-((S3*S3)/X9)
1890 S2=B4-((B3*B3)/X9)
1900 R1=SQR(S2/S1)
1910 @ : @ : @"LINEAR REGRESSION:"
1920 @
1930 @USING"-#######.###","Y =";S4;" +";S5;" X"
1940 @USING"-##.###","R  ";R1
1950 @
1960 INPUT"OO YOU WANT TO DELETE ANALYSIS IN REGRESSION? (YES OR NO):  ",V$
1970 IF V$="YES"THEN GOSUB 2000
1980 IF V$="YES"THEN GOTO 1690
1990 RETURN
2000 REM SUBR TO DELETE ANALYSIS IN REGRESSION
2010 INPUT"OO YOU NEED THE FULL ANALYSIS PRINTED OUT?  (YES OR NO):  ",W$
2020 IF W$="NO"THEN GOTO 2040
2030 GOSUB 2260
2040 INPUT"HOW MANY ANALYSIS TO DELETE (MAX 20):  ",W
2050 DIM C(W)
2060 @"GIVE THE NUMBERS ON THE ANALYSES TO DELETE IN THE DATA FILE"
2070 @"WHICH STARTS WITH NO. 1 (e.g. ?2 ?5 ?15 ?55)."
2080 L=0
2090  FOR I=0 TO W-1
2100  IF L=W-1 THEN@"AND THE LAST ANALYSES TO DELETE - "
2110  INPUT C(I)
2120  L=L+1
2130  NEXT I
2140 C=0
2150  FOR I=0 TO W-1
2160   FOR L=C(I)-1-C TO X9-1
2170   IF L=X9-1 THEN GOTO 2200
2180   A(0,L)=A(0,L+1) : A(1,L)=A(1,L+1)
2190   NEXT L
2200  C=C+1
2210  NEXT I
2220 X9=X9-W
2230 INPUT"OO YOU INTEND TO PLOT THE NEW OATA FILE?  (YES OR NO):  ",Z$
2240 IF Z$="YES"THEN GOSUB 2360
2250 RETURN
2260 REM SUBR TO PRINT A-ARRAY
2270 S=1 : @
2280  FOR I=0 TO X9-1
2290  @"  NO. ";S;"   ";A(0,I);"-";A(1,I);"       ";
2300  IF INT(S/2)=S/2 THEN@
2310  S=S+1
```

```
2320  NEXT I
2330 @ : @ : @
2340 RETURN
2350 STOP
2360 REM SUBR TO REPLACE OLD WITH NEW DATA FILE
2370  FOR I=0 TO X9-1
2380  J(I)=A(0,I) : T(I)=A(1,I)
2390  NEXT I
2400 N=X9-1
2410 RETURN
```

Appendix 5. Listing of HISTOGR

```
>>LOAD"HISTOGR"

>>LIST

 100 REM PROGRAM TO PRINT HISTOGRAM. DATA READ FROM RECORD FILE.
 110 DIM P$(40)
 120  FOR I=0 TO 70
 130  @"#"; : NEXT I : @
 140 @ : @ : @"HISTOGRAM PLOTTING PROGRAM.     VER. 2."
 150 @"NORDIC VOLCANOLOGICAL INSTITUTE. 26.7.80.  PT."
 160 INPUT"GIVE DATA FILE:  ",S$
 170 OPEN\1,120\S$
 180 GET\1,0\K1,P$
 190 @ : @ : @
 200 INPUT"GIVE NUMBER OF ANALYSIS?  ",K1
 210 DIM C(K1),O(200),E(100)
 220 K1=K1-1
 230 H1=0 : F8=0 : F9=100
 240  FOR J=0 TO K1
 250  H1=H1+1
 260  GET\1,H1\C(J)
 270  NEXT J
 280 INPUT"GIVE MAX, MIN  (0,0 CAUSES DEFAULT): ",F8,F9
 290 IF F8=0 AND F9=0 THEN GOSUB 740
 300 @"MAX :";F8;"  MIN :";F9
 310 INPUT"GIVE STEP IN HISTOGRAM?  ",L1
 320 G1=(F8-F9)/L1
 330 IF INT(G1)<>G1 THEN F8=(INT(G1)*L1)+L1+F9
 340 IF INT(G1)<>G1 THEN@"MAX CHANGED TO :";F8
 350 Y1=0
 360  FOR I=0 TO 200
 370  O(I)=0 : NEXT I
 380  FOR I=0 TO 100
 390  E(I)=0 : NEXT I
 400  FOR I=F9 TO F8 STEP L1
 410  Y1=Y1+1
 420  O(Y1)=I
 430  NEXT I
 440  FOR J=0 TO(G1+1)
 450   FOR I=0 TO K1
 460   IF C(I)<=O(J)+(0.5*L1)AND C(I)>O(J)-(0.5*L1)THEN E(J)=E(J)+1
 470   NEXT I
 480  NEXT J
 490  FOR I=0 TO 4
 500  @
 510  NEXT I
 520  FOR I=1 TO(G1+1)
 530  @USING"###.###",O(I);
 540  @" !";
 550  IF E(I)=0 THEN GOTO 600
 560   FOR J=0 TO E(I)
 570   IF E(I)=J THEN GOTO 600
 580   @"*";
 590   NEXT J
 600  @
 610  NEXT I
 620 @ : @
 630 @"N=";" ";H1
 640 @ : @P$
 650  FOR I=0 TO 4
 660  @
 670  NEXT I
 680 INPUT"DO YOU WANT TO TERMINATE THE PROGRAM?  (YES OR NO):  ",Z$
 690 IF Z$="YES"THEN GOTO 710
 700 GOTO 310
 710 @"PROGRAM TERMINATED."
 720 CLOSE\1\
 730 STOP
 740 REM SUBR TO FIND MAX AND MIN
 750 F8=C(0) : F9=C(0)
 760  FOR I=0 TO K1
 770  F8=MAX(F8,C(I))
 780  F9=MIN(F9,C(I))
 790  NEXT I
 800 RETURN
```